Appendix 1

Mplus Source Code for Data Simulation and Fitting of Model Used in the Illustration Section

```
TITLE:              SIMULATING THE DATA SETS USED IN THE ILLUSTRATION SECTION.

                    (STEP 1/CODE 1; SEE EQUATIONS (1) AND (8) FOR DETAILS).

MONTECARLO:         NAMES = Y1-Y6; ! k = 6 scale components.

                    NGROUPS = 2; ! g = 2 groups.

                    NOBS = 800 800; ! group sample size n = 800.

                    NREPS = 10000; ! number of requested replications = 10,000.

                    SEED = 220316; ! use this seed to replicate all results.

                    REPSAVE = ALL; ! save all 10,000 replication data sets.

                    SAVE = PMI_MR*.DAT; ! name of overall data file with reps.

MODEL POPULATION:

                    F BY Y1*1 Y2*1.5 Y3-Y6*2; ! see Equations (8) and their

                    F@1; ! following discussion for all these population

                    Y1-Y6*2; ! parameters that are used in the simulation.

                    [F@0]; ! latent mean fixed at 0 in group 1 for identif.

                    [Y1-Y6@0]; ! intercepts selected as 0 in group 1

MODEL POPULATION-G2:

                    F BY Y1*1 Y2*1.5 Y3-Y5*2 Y6*1; ! model in group 2

                    F@1.333 ! 33% of the latent variance in group 1 is built as

                            ! group difference in the latent variances in the

                            ! simulation process.
```

```
                    Y1-Y5*2;  ! same error variances apart from last component.
                    Y6*.5; ! group unequal error variance (see main text).
                    [F@.333]; ! also 33% of the latent variance in group 1 is
                              ! built as group difference in the latent
                              ! variances in the simulation process.
                    [Y1-Y6@0]; ! intercepts selected as 0 also in group 2.
MODEL:
                    F BY Y1-Y6; ! this model is irrelevant, and stated here
                     ! only i.o. to 'close' the data simulation process at the
                     ! software level.
OUTPUT:             TECH9; ! check last output section for possible error
                           ! indications or related warnings, if any.
```

Note. Exclamation mark is used to start an annotating comment stated in the rest of the command line. (The 10,000 simulated replication data sets contain by default as last column the group membership variable; see also next command file.)

```
TITLE:      FITTING THE TWO-GROUP MODEL USED, TO THE 10,000 REPLICATION
            SIMULATED DATA SETS, AND EVALUATION OF MR AND RELATED QUANTITIES.
            (STEP 2/CODE 2; see main text).
DATA:       FILE = PMI_MRLIST.DAT; ! Use file with all 10,000 replications.
            TYPE = MONTECARLO; ! Requests successive selection of the
VARIABLE:   NAMES = Y1-Y6 GROUP; ! 10,000 data sets/reps., one by one.
            GROUPING = GROUP(1=G1, 2=G2); ! Setting up a two-group model.
```

```
MODEL:
            F BY Y1*(B1) ! Assign parameter labels used further below.
             Y2-Y6* (B2-B6);
            F@1; [F@0]; ! Group 1 constraints for model identification.
            Y1-Y6(TH11-TH16);
MODEL G2:
            F BY Y1*(B1)
             Y2-Y5* (B2-B5)
             Y6*(B26); ! The last component loading is not group invariant.
            F*1(LV_G2); ! The latent variance is free in group 2.
            [F*]; ! The latent mean is free in group 2.
            Y1-Y6(TH21-TH26);
MODEL CONSTRAINT:
            NEW(MR1, MR2, MR1_NUM, MR2_NUM, DELTA);
            MR1_NUM = B1^2/TH11+B2^2/TH12+B3^2/TH13+
             B4^2/TH14+B5^2/TH15+B6^2/TH16; ! numerator of MR in group 1
            MR1 = MR1_NUM/(1+MR1_NUM); max. rel. coefficient in group 1
            MR2_NUM = B1^2/TH21+B2^2/TH22+B3^2/TH23+
             B4^2/TH24+B5^2/TH25+B26^2/TH26; ! numerator of MR in group 2
            MR2 = MR2_NUM/(1+MR2_NUM); max. rel. coefficient in group 2
            DELTA = MR1-MR2; ! GDCP index, Δ, defined in Equation (5).
SAVEDATA:   RESULTS = PMI_MR_RESULTS_STEP2.DAT; ! all parameter estimates are
             ! found in this results file (the latent mean and variance
```

```
              ! estimates are in columns 27 and 28).
OUTPUT:     TECH1 CINTERVAL; ! need the latter if model fitted to a single
              ! data set, for interval estimation of the GDCP index (5) then.
```

Note. When this command file is used on a single data set, the 90%-, 95%-, and 99%-confidence intervals of the Δ index (5) will be provided in the output section titled "Confidence intervals of model parameters".

Appendix 2

Reliability, Linear Combinations of Measuring Instrument Components, and Empirical Evaluation of the GDCP Index

Reliability and linear combinations

As is well-known, the reliability coefficient of an observed measure equals the R-square index of the conceptual regression of its associated latent score upon the observed score (or conversely; e.g., McDonald, 1999). For a given population and measuring instrument (such as a scale, composite, self-report, survey, inventory, or questionnaire), and under the assumptions of the model defined in Equations (1), the reliability coefficient of any linear combination Z of the instrument components that is defined as

(A2.1)
$$Z = w_1 y_1 + \ldots + w_k y_k =$$
$$w_1 \alpha_1 + \ldots + w_k \alpha_k + (w_1 b_1 + \ldots + w_k b_k)f + w_1 e_1 + \ldots + w_k e_k$$

with the w's being respective weights, is the R-square index of the conceptual regression of Z on what may be treated as its trait, construct or factor score $f_Z = w_1\alpha_1 + \ldots + w_k\alpha_k + (w_1b_1 + \ldots + w_kb_k)f$. (The group subindex is dropped here and in the remainder of this Appendix.)

We next observe that this R-square is identical to that index when regressing conversely f on Z, which is a useful conceptual regression for the goals of this discussion (cf. Raykov & Marcoulides, 2018). Therefore, keeping in mind that the regression approach is directly related to prediction of a response by an explanatory variable (e.g., Agresti & Finlay, 2009), it is readily noticed that the latter R-square index can also be viewed as a coefficient of predictability of f by Z. It is worth mentioning that while these two regressions cannot be carried out in an empirical study, due to the latent variable f not being associated with any observations, they provide a conceptual link to the aims of this article which we will capitalize on in the sequel.

We then note that the R-square index associated with the regression of f on Z, depends on the linear combination weights $w_1, \ldots, w_k$ in Equation (2). In particular, when all of them are 1, that combination Z represents the simple overall sum score that is traditionally and widely used in empirical research, which for convenience we denote as Y, i.e.,

(A2.2) $$Y = y_1 + \ldots + y_k \, .$$

In that case, the R-square of the conceptual regression of f on Y is the popular scale reliability coefficient (e.g., McDonald, 1999). We symbolize this coefficient by r_Y, and as usual define it as the proportion of true variance in observed variance in Y (e.g., Raykov & Marcoulides, 2011), which proportion can be estimated by that of observed variance explained by the underlying factor, construct or studied trait in the setting of interest in this paper; see Equation (1), for a given group; see also Bollen, 1980).

The preceding discussion in this Appendix prompts the question if the w-weights in Equation (A2.1) can be selected in such a way that the resulting R-square index, for the regression of f on Z, exceeds the scale reliability r_Y. If such a choice was possible, it would follow that the above linear combination Z with those alternative weights would be superior to the simple overall sum score Y in terms of measurement consistency. This question of increasing reliability by using an appropriate linear combination of the components of a given measuring instrument, has attracted a great deal of interest since the 1940s. The query is responded to by the concepts of MR and associated OLC, which are discussed in the main text.

Point and interval estimation of the GDCP index

Based on the parametric representation (6) of the GDCP index Δ, we can make the following observation with respect to its estimation in an empirical social measurement setting. Specifically, when model (1) is plausible in a two-group study, point and interval estimation of the GDCP index (5) is straight-forwardly achievable with the popular LVM methodology and widely circulated LVM software Mplus (Muthén & Muthén, 2024; for pertinent details, see Code 2 in Appendix 1). For example, with component normality, a maximum likelihood (ML) point estimate of the GDCP index (5) is directly obtained by substituting the ML factor loading and error variance estimates furnished when fitting this model using ML. The frequently utilized delta-method in applications, also implemented in the software, can then be readily employed for obtaining a confidence interval of the GDCP index Δ at any pre-specified confidence level (e.g., Raykov & Marcoulides, 2004; see Appendix 1 and Note to the second Mplus command file in it). Alternatively, with up to mild violations of normality (or in case with items possessing at least 5-7 possible response options and not excessively skewed distributions), the robust ML can be used for model testing and interval evaluation of the index Δ (Rhemtulla, Brosseau-Liard, & Savalei, 2012; Muthén & Muthén, 2024; see also discussion and conclusion section).